PharmaSUG 2017 - Paper BB15

A Case of Assessing Adverse Events of Interest Based on Their Grade Changes

Sreeram Kundoor, Seattle Genetics, Inc., Bothell, WA

ABSTRACT:

Adverse Events play an important role in assessing the safety of a drug. While shift tables are commonly used to describe laboratory data and determine shifts in toxicity grades, a similar approach can also be taken to determine the AE grade shifts in describing AE data. These shift tables provide an insight on whether a subject is recovering/resolving or recovered/resolved and the KM analysis provides further insight of the median duration for the improvement or resolution which in turn helps the safety assessment of the product. In order to do this, we have to filter the data for the adverse events of special interest and find the shift from chosen (Maximum) grade to the subsequent lowest grade.

With suitable examples, this paper describes an approach to create AE shift tables and the KM estimate for the median duration of improvement or resolution. This paper also discusses various challenges and complex scenarios encountered and an approach to overcome the issues (and the logic behind them).

INTRODUCTION:

Shift tables are commonly used to determine laboratory toxicity grade shifts and assessing safety. We took a similar approach in assessing the grade shifts of adverse events of interest starting with identifying the adverse events of special interest based on the System organ class and determining the shift of first occurrence of maximum severity events to subsequent lowest grade of grade 2 or better or resolution. This gives an insight into how low the grades are shifting after a subject had an event of higher intensity and their outcomes. Once the shifts have been determined, we can calculate the duration of the events from first occurrence of grade 3 or 4 to their subsequent lowest grade of 2 or better and accordingly determine the KM estimates of the duration. If the event outcome is “Recovered/Resolved”, then the shift is considered to be a grade 0. This analysis helps in identifying the pattern of the adverse events and their shift from the maximum grade to the subsequent lowest grade or if the events are recovering or resolving and if there are any event free days. Although we know that these are treatment emergent adverse events, this attempt gives us the scope to further analyze if the events are getting resolved or worsened and the duration it takes for the condition to either get better or worsen. If we extend this further, we can also analyze if the adverse events are occurring due to the accumulation of the study drug or if the intensity of the events is being reduced during the resting period, which might give a substantial evidence whether the grade 3 or 4 events will most likely recover or resolve once the dosing is stopped.

APPROACH:

- Select adverse events of interest
- Consolidate all the events based on the grade and start date.
- Pick the first occurrence of the maximum grade event
- Determine the subsequent lowest grade events
- Determine the duration from start of the grade 3/4 event to subsequent lowest grade 2 or better events.

SELECTING THE AE'S OF INTEREST AND CONSOLIDATING THEM BASED ON THE GRADE

Select the treatment emergent adverse events of interest and assign the grade as "0" for events with outcome of 'RECOVERED/RESOLVED','RECOVERED/RESOLVED WITH SEQUELAE' and then determine the maximum toxicity grade and AE start date of the first occurrence of the maximum severity. The code below is used to determine the first occurrence of maximum severity grade.

```
*Subset of treatment emergent AE's of interest;

data ae1;
      set a.adae;
      where aetrtem='Y' and AE_INT='Y';
      aetoxgrn=input(aetoxgr, best.);
        if aeout in ('RECOVERED/RESOLVED', 'RECOVERED/RESOLVED WITH
SEQUELAE') then tempgrd=0;
         else tempgrd=aetoxgrn;
run;

proc sort data=ae1;
      by usubjid aestdt aeendt descending aetoxgrn;
run;

*Getting maximum grade and maximum grade date;
proc sql;
      create table max as
      select distinct usubjid, max(aetoxgrn) as maxgrd, min(aestdt) as
      maxgrddt format=date9.
      from adae
      group by usubjid, aetoxgr;
quit;

data max;
      set max;
      by usubjid ;
      if last.usubjid;
run;
```

After determining the Maximum grade, merge this dataset with the original AE dataset and determine the subsequent lowest grade. It should be noted that each subject and events occurred to them are unique

and differs case by case. We need to look if there are any events which occurred before the maximum severity grade and are they still ongoing or if the subject has only one single event and that has been recovered or resolved or if the subject is event free for at least one single day.

In order to handle these different scenarios, we took a block by block approach, where subjects who fall under each category are dealt in a different way.

A) Handling subjects with ongoing events occurred prior to max toxicity grade

As discussed above, each category of the patients should be handled in a different programmatic manner. The code shown below is used to select the subjects who have an event with lower toxicity grade than the maximum grade and the event was not recovered or resolved. This means, it is an ongoing event which occurred before the maximum severity grade, so these events are flagged as ongoing events. A minimum grade is to be determined for the rest of the subjects who don't fall under this particular category.

```
*Merging with the original AE data;
data adae2 aeltmax;
      merge adae max(in=a);
      by usubjid ;
      if a;
      if aestdt=. then aestdt=iaestdt;
      *subjects with ongoing events prior to max grade;
      if aestdt<=maxgrddt and aeout in ("NOT RECOVERED/NOT RESOLVED") and
      aeendt=. then output aeltmax;
      else output adae2;
run;

*Determining the first occurrence of minimum date;
data min;
      set adae2;
      by usubjid aetoxgrn aestdt;
      if first.usubjid;
      mingrd=aetoxgrn;
      mingrddt=aestdt;
      minendt=aeendt;
      minout=aeout;
      where aestdt ne .;
      format mingrddt minendt date9.;
      keep usubjid mingrd mingrddt minendt tempgrd minout;
run;
```

B) Handling subjects with one single event and which is recovered and resolved

In this particular case, there are subjects who only have one single event and for these subjects, if the event is recovered or resolved, then the minimum grade would be '0', else there won't be any grade change. ***Merging back the minimum grade and date information to original data;

```
data all;
      merge adae2 min;
      by usubjid;
run;
```

```
data adae_a adae_b;
      set all;
      by usubjid;
      if first.usubjid and last.usubjid then output adae_a;
      else output adae_b;
run;

data adae_a;
      set adae_a;
      mingrd=tempgrd;
      keep usubjid maxgrd mingrd;
run;
```

c) Handling subjects with at least one single event free day

The concept of this shift table is to see if a subject had any event free day or if the severity is decreased from the maximum severity grade. If the subject does not have any event for a certain period of time, the grade shift is considered as recovered/resolved or a grade '0' from the maximum severity.

```
data adae_b1;
      set adae_b;
      if aeendt=. then aeendt=opthdt;
      do while (aestdt <= aeendt );
            output;
            aestdt = aestdt + 1;
      end;
run;
data adae_b2 adae_b3;
      set adae_b1;
      by usubjid;
      lagstdt=lag(aestdt);
      if first.usubjid then do;
            lagstdt=.;
            lagendt=.;
      end;
      daydiff=aestdt-lagstdt;
      format lagstdt date9.;
      if daydiff>1 then output adae_b2;
      else output adae_b3;
run;
```

D) Handling subjects with adverse event outcome of 'Recovered/ Resolved'

These are the subjects who had a subsequent lowest grade or the severity grade equivalent to the maximum grade and the outcome is recovered/resolved. In this case, as they are recovered/resolved, the grade change must be '0' from the maximum grade.

```
data adae_b3;
      set adae_b3;
      by usubjid;
      if last.usubjid and aeout in ("RECOVERED/RESOLVED","RECOVERED/RESOLVED
      WITH SEQUELAE") then aetoxgrn=tempgrd;
run;
```

E) Handling subjects with adverse event outcome of 'Not Recovered/ Not Resolved'

If a subject is not recovered or not resolved after the maximum toxicity grade, then the subsequent lowest toxicity grade would be the event with lower or similar intensity occurring after the maximum toxicity grade.

```
data adae_b3 adae_b33;
      set adae_b3;
      if mingrd=maxgrd and aeout="NOT RECOVERED/NOT RESOLVED" then output
      adae_b33;
      else output adae_b3;
run;

data adae_b3c;
      set adae_b33;
      by usubjid;
      if last.usubjid;
      mingrd=aetoxgrn;
      keep usubjid maxgrd mingrd  ;
run;
```

F) All the rest of the subjects and their maximum and Minimum grades

```
proc sql;
      create table adae_b3aa as
      select distinct usubjid, maxgrd, min(aetoxgrn) as mingrd
      from adae_b3
      group by usubjid;
quit;
```

G) Final dataset by merging all the intermediate datasets

```
data adae_b2;
      merge adae_b3aa adae_b3c adae_b33 adae_b2;
      by usubjid;
run;


data final;
      set adae_a adae_b2;
      proc sort nodupkey; by usubjid;
run;
```

This final dataset contains all the subjects with adverse events of interest and their maximum and subsequent lowest grades. An adverse event shift table can be generated by determining the frequency counts between the maximum grade and minimum grade variables.

Shift from Maximum Severity to Subsequent Lowest Severity
All Treated Patients Set

	Subsequent Lowest Severity					
Maximum Severity	0[a] (N=18)	1 (N=14)	2 (N=20)	3 (N=3)	4 (N=0)	Total (N=55)
1	0	1	0	0	0	1
2	4	4	9	0	0	17
3	14	5	9	3	0	31
4	0	4	2	0	0	6

Figure 1: Shift from Maximum grade to subsequent lowest grade

KAPLAN-MEIER ANALYSIS OF GRADE 3 OR 4 EVENTS TO 2 OR BETTER OR RESOLUTION

After determining the grade shifts, select the maximum grade 3 or 4 events along with the event start date and determine the subsequent lowest grade 2 or better events and the event end date, based on the start and end dates of the events calculate the duration of the events. If a subject has an ongoing event even after the treatment end date, then the last examination or event evaluation date is to be considered as the end date, the subject is censored on that date and Kaplan-Meier estimates are to be performed based on the duration and censor variable. This gives the median duration of the events per treatment group and their 95% confidence intervals.

Grade 3 or 4 Adverse Events of interest to Grade 2 or Better
All Treated Patients Set

	X mg/kg (N=3) n (%)	X+1 mg/kg (N=3) n (%)	X+2 mg/kg (N=2) n (%)	X+3 mg/kg (N=15) n (%)	X+4 mg/kg (N=15) n (%)	X+5 mg/kg (N=13) n (%)	X+6 mg/kg (N=4) n (%)	Total (N=55) n (%)
Number of patients with the Adverse Events of interest, n(%)	0	0	1 (50)	12 (67)	8 (53)	8 (45)	3 (75)	32 (52)
Number of patients with improvement or resolution	0	0	0	11 (91)	6 (75)	8 (100)	3 (100)	28 (88)
Number of patients without improvement or resolution	0	0	1 (100)	1 (9)	2 (25)	0	0	4 (12)
Median duration of the event [c] (weeks) (95% CI [d])	- (-, -)	- (-, -)	- (-, -)	3.3 (1.1, 5.4)	7.9 (2.4, 21.6)	5.1 (4.0, 16.1)	3.6 (1.1, 7.1)	5.0 (3.1, 6.3)
25th, 75th percentile	-, -	-, -	-, -	3.1, 5.0	2.6, 17.1	4.1, 14.1	1.1, 7.1	3.1, 7.0
Observed min, max	-, -	-, -	58.0, 58.0	1.1, 6.3	0.1, 21.6	4.0, 16.1	1.1, 7.1	0.1, 58.0

Figure 2: Kaplan Meier analysis of grade 3 or 4 adverse events to grade 2 or better.

CONCLUSION:

The analysis and the approach will give an insight on how the grades are shifting and the KM analysis provides the median and the 95% confidence intervals of the duration of the events. This analysis would be very helpful in supporting the safety evaluation of the compound and making a statement on whether the toxicities associated with the compound would get better or worsen with time and the duration of the events.

CONTACT INFORMATION:

Your comments and questions are valued and encouraged. Contact the author at:

Name: Sreeram Kundoor (Sriramu Kundoor)
Enterprise: Seattle Genetics Inc.
Address: 21823 30th Drive SE
City, State ZIP: Bothell, WA 98021
E-mail: skundoor@seagen.com
Web: http://www.seattlegenetics.com/